
Agenda for Today
n Hardware Description Languages

n Implementing Combinational Logic (in Verilog)

n Implementing Sequential Logic (in Verilog)

What We Will Cover Soon: LC-3 Processor

What We Will Cover Soon: LC-3 Datapath

Readings (This Week)
n Hardware Description Languages and Verilog 

q H&H Chapter 4 in full

n Timing and Verification
q H&H Chapters 2.9 and 3.5 + (start Chapter 5)

n By tomorrow, make sure you are done with 
q P&P Chapters 1-3    +      H&H Chapters 1-4

Readings (Next Week)
n Von Neumann Model, LC-3, and MIPS

q P&P, Chapter 4, 5
q H&H, Chapter 6
q P&P, Appendices A and C (ISA and microarchitecture of LC-3)
q H&H, Appendix B (MIPS instructions)

n Programming
q P&P, Chapter 6

n Recommended: Digital Building Blocks
q H&H, Chapter 5

Hardware Description Languages 
& Verilog

2017: Intel Kaby Lake

• 64-bit processor
• 4 cores, 8 threads
• 14-19 stage pipeline
• 3.9 GHz clock freq.

• 1.75B transistors

• In ~47 years, about 
1,000,000-fold 
growth in transistor 
count and 
performance! 

https://en.wikichip.org/wiki/intel/microarchitectures/kaby_lake

https://en.wikichip.org/wiki/intel/microarchitectures/kaby_lake


2021: Apple M1

9Source: https://www.anandtech.com/show/16252/mac-mini-apple-m1-tested 

• 4 High-Perf GP Cores
• 4 Efficient GP Cores
• 8-Core GPU
• 16-Core Neural 

Engine
• Lots of Cache
• Many Caches
• 8x Memory Channels

• 16B transistors

https://www.anandtech.com/show/16252/mac-mini-apple-m1-tested


2022: Apple M1 Ultra

10https://www.theverge.com/2022/3/9/22968611/apple-m1-ultra-gpu-nvidia-rtx-3090-comparison

• 16 High-Perf GP Cores
• 4 Efficient GP Cores
• 64-Core GPU
• 32-Core Neural Engine
• Lots of Cache
• Many Caches
• 32x Memory Channels
• 128 GB DRAM

• 114B transistors



2019: Cerebras Wafer Scale Engine

Cerebras WSE               
1.2 Trillion transistors

46,225 mm2   

Largest GPU               
21.1 Billion transistors

815 mm2   

n The largest ML 
    accelerator chip (2019)

n 400,000 cores 

NVIDIA TITAN V
https://www.anandtech.com/show/14758/hot-chips-31-live-blogs-cerebras-wafer-scale-deep-learning

https://www.cerebras.net/cerebras-wafer-scale-engine-why-we-need-big-chips-for-deep-learning/

https://www.anandtech.com/show/14758/hot-chips-31-live-blogs-cerebras-wafer-scale-deep-learning
https://www.cerebras.net/cerebras-wafer-scale-engine-why-we-need-big-chips-for-deep-learning/


2021: Cerebras Wafer Scale Engine 2

Cerebras WSE-2               
2.6 Trillion transistors

46,225 mm2   

Largest GPU               
54.2 Billion transistors

826 mm2   

n The largest ML 
    accelerator chip (2021)

n 850,000 cores 

NVIDIA Ampere GA100
https://www.anandtech.com/show/14758/hot-chips-31-live-blogs-cerebras-wafer-scale-deep-learning

https://www.cerebras.net/cerebras-wafer-scale-engine-why-we-need-big-chips-for-deep-learning/

https://www.anandtech.com/show/14758/hot-chips-31-live-blogs-cerebras-wafer-scale-deep-learning
https://www.cerebras.net/cerebras-wafer-scale-engine-why-we-need-big-chips-for-deep-learning/


Transistor Counts Are Growing

Memory chips have orders of magnitude more transistors than computation chips

Source: https://en.wikipedia.org/wiki/Transistor_count 

https://en.wikipedia.org/wiki/Transistor_count


How to Deal with This Complexity?
n Hardware Description Languages

n What we need for hardware design:
q Ability to describe (specify) complex designs 
q … and to simulate their behavior (functional & timing)
q … and to synthesize (automatically design) portions of it

n have an error-free path to implementation

n Hardware Description Languages enable all of the above
q Languages designed to describe hardware
q There are similarly-featured HDLs (e.g., Verilog, VHDL, ...)

n if you learn one, it is not hard to learn another
n mapping between languages is typically mechanical, especially for 

the commonly used subset

Hardware Description Languages
n Two well-known hardware description languages

n Verilog
q Developed in 1984 by Gateway Design Automation
q Became an IEEE standard (1364) in 1995
q More popular in US

n VHDL (VHSIC Hardware Description Language)
q Developed in 1981 by the US Department of Defense
q Became an IEEE standard (1076) in 1987
q More popular in Europe

n We will use Verilog in this course

Why Specialized Languages for Hardware?

n HDLs enable easy description of hardware structures
q Wires, gates, registers, flip-flops, clock, rising/falling edge, …
q Combinational and sequential logic elements

n HDLs enable seamless expression of parallelism inherent in 
hardware
q All hardware logic operates concurrently

n Both of the above ease specification, simulation & 
synthesis

Hardware Design Using HDL

Key Design Principle: Hierarchical Design
n Design a hierarchy of modules

q Predefined “primitive” gates (AND, OR, …) 
q Simple modules are built by instantiating 

these gates (e.g., components like MUXes)
q Complex modules are built by instantiating 

simple modules, …

n Hierarchy controls complexity 
q Analogous to the use of function/method 

abstraction in programming

n Complexity is a BIG deal
q In real world, how big is the size of a 

module (that is described in HDL and then 
synthesized to gates)?

How many?

https://techreport.com/review/21987/intel
-core-i7-3960x-processor

https://techreport.com/review/21987/intel-core-i7-3960x-processor
https://techreport.com/review/21987/intel-core-i7-3960x-processor


Top-Down Design Methodology
n We define the top-level module and identify the 

sub-modules necessary to build the top-level module
n Subdivide the sub-modules until we come to leaf cells

q Leaf cell: circuit components that cannot further be divided 
(e.g., logic gates, primitive cell library elements)

Top-level 
Module

Sub-module Sub-module Sub-module

… … ……
Leaf-cell Leaf-cell Leaf-cell Leaf-cell



Bottom-Up Design Methodology

20

n We first identify the building blocks that are available to us
n Build bigger modules, using these building blocks
n These modules are then used for higher-level modules until 

we build the top-level module in the design

Top-level 
Module

Sub-module Sub-module Sub-module
… … ……

Leaf-cell Leaf-cell Leaf-cell Leaf-cell



Defining a Module in Verilog
n A module is the main building block in Verilog

n We first need to define:
q Name of the module
q Directions of its ports (e.g., input, output)
q Names of its ports

n Then:
q Describe the functionality of the module

a
b y
c

Verilog
Module

21

inputs output

example



Implementing a Module in Verilog

a
b y
c

Verilog
Module

module example (a, b, c, y);
 input a;
 input b; 
 input c;
 output y;

// here comes the circuit description

endmodule

22

Port list
(inputs and outputs)

ports have a 
declared type

a module 
definition

name of 
module

example



n The following two codes are functionally identical

A Question of Style (and Consistency)

module test ( a, b, y );
 input a;
 input b;
 output y;

endmodule

module test ( input a,
              input b,
              output y );

endmodule

23

port name and direction declaration 
can be combined



What If We Have Multi-bit Input/Output?

n You can also define multi-bit Input/Output (Bus) 
q [range_end : range_start]
q Number of bits: range_end – range_start + 1

n Example:

n a represents a 32-bit value, so we prefer to define it as: 
[31:0] a

n It is preferred over [0:31] a which resembles array definition
n It is good practice to be consistent with the representation 

of multi-bit signals, i.e., always [31:0] or always [0:31]

input  [31:0] a;   // a[31], a[30] .. a[0]
output [15:8] b1;  // b1[15], b1[14] .. b1[8]
output [7:0]  b2;  // b2[7], b2[6] .. b2[0]
input         c;   // single signal

Manipulating Bits

// You can assign partial buses
   wire [15:0] longbus;
   wire [7:0] shortbus;
   assign shortbus = longbus[12:5];

// Concatenating is by {}
   assign y = {a[2],a[1],a[0],a[0]};

 
// Possible to define multiple copies
   assign x = {a[0], a[0], a[0], a[0]}
   assign y = { 4{a[0]} }

25

n Bit Slicing
n Concatenation
n Duplication



Basic Syntax
n Verilog is case sensitive

q SomeName and somename are not the same!
n Names cannot start with numbers:

q 2good is not a valid name
n Whitespaces are ignored

// Single line comments start with a //

/* Multiline comments 
   are defined like this */

Two Main Styles of HDL Implementation
n Structural (Gate-Level)

q The module body contains gate-level description of the circuit
q Describe how modules are interconnected
q Each module contains other modules (instances)
q … and interconnections between those modules
q Describes a hierarchy of modules defined as gates

n Behavioral
q The module body contains functional description of the circuit
q Contains logical and mathematical operators
q Level of abstraction is higher than gate-level

n Many possible gate-level realizations of a behavioral description

n Many practical designs use a combination of both
27



Structural (Gate-Level) HDL

28



Structural HDL: Instantiating a Module

29

Schematic of module “top” that is built from 
two instances of module “small”

i_first
i_second



Structural HDL Example
n Module Definitions in Verilog

module top (A, SEL, C, Y);
  input A, SEL, C;
  output Y;
  wire n1;  

endmodule

module small (A, B, Y);
  input A;
  input B;
  output Y;

// description of small

endmodule 

30

i_first
i_second



module top (A, SEL, C, Y);
  input A, SEL, C;
  output Y;
  wire n1;  

endmodule

module small (A, B, Y);
  input A;
  input B;
  output Y;

// description of small

endmodule 

Structural HDL Example
n Defining wires (module interconnections)

31

i_first
i_second



n The first instantiation of the “small” module
module top (A, SEL, C, Y);
  input A, SEL, C;
  output Y;
  wire n1;

// instantiate small once
small i_first ( .A(A),
                .B(SEL),
                .Y(n1)   );
  

endmodule

module small (A, B, Y);
  input A;
  input B;
  output Y;

// description of small

endmodule 

Structural HDL Example

32

i_first
i_second



n The second instantiation of the “small” module

module top (A, SEL, C, Y);
  input A, SEL, C;
  output Y;
  wire n1;  

// instantiate small once
small i_first ( .A(A),
                .B(SEL),
                .Y(n1)   );

// instantiate small second time
small i_second ( .A(n1),
           .B(C),
           .Y(Y) );

endmodule

module small (A, B, Y);
  input A;
  input B;
  output Y;

// description of small

endmodule 

Structural HDL Example

33

i_first
i_second



n Short form of module instantiation

module top (A, SEL, C, Y);
  input A, SEL, C;
  output Y;
  wire n1;  

// alternative short form
small i_first ( A, SEL, n1 );

/* In the short form above,
   pin order very important */ 

// safer choice; any pin order
small i_second ( .B(C),
           .Y(Y),
           .A(n1) );

endmodule

module small (A, B, Y);
  input A;
  input B;
  output Y;

// description of small

endmodule 

Structural HDL Example

i_first
i_second

Short form is not good practice 
as it reduces code maintainability



Structural HDL Example (II)
n Verilog supports basic logic gates as predefined primitives

q These primitives are instantiated like modules except that they 
are predefined in Verilog and do not need a module definition

35

module mux2(input d0, d1, 
            input s,
            output y);
   wire ns, y1, y2;
   
   not  g1 (ns, s);
   and  g2 (y1, d0, ns);
   and  g3 (y2, d1, s);
   or   g4 (y, y1, y2);
   
endmodule

A B

S

C

ba

d0 d1

s

y

y1 y2



Behavioral HDL

Recall: Two Main Styles of HDL Implementation

n Structural (Gate-Level)
q The module body contains gate-level description of the circuit
q Describe how modules are interconnected
q Each module contains other modules (instances)
q … and interconnections between those modules
q Describes a hierarchy of modules defined as gates

n Behavioral
q The module body contains functional description of the circuit
q Contains logical and mathematical operators
q Level of abstraction is higher than gate-level

n Many possible gate-level realizations of a behavioral description

n Many practical designs use a combination of both
37



Behavioral HDL: Defining Functionality

module example (a, b, c, y);
 input a;
 input b; 
 input c;
 output y;

// here comes the circuit description
assign y = ~a & ~b & ~c |
            a & ~b & ~c |
            a & ~b &  c;   

endmodule

Behavioral HDL: Schematic View

39

A behavioral implementation still models a 
hardware circuit!

ANDa

y
b

c

AND

AND

OR



Bitwise Operators in Behavioral Verilog

module gates(input  [3:0]  a, b,
             output [3:0] y1, y2, y3, y4, y5);

   /* Five different two-input logic 
      gates acting on 4 bit buses */

   assign y1 = a & b;    // AND
   assign y2 = a | b;    // OR
   assign y3 = a ^ b;    // XOR
   assign y4 = ~(a & b); // NAND
   assign y5 = ~(a | b); // NOR

endmodule

Bitwise Operators: Schematic View

Reduction Operators in Behavioral Verilog

module and8(input  [7:0] a, 
            output       y);

   assign y = &a;

   // &a is much easier to write than
   // assign y = a[7] & a[6] & a[5] & a[4] &
   //            a[3] & a[2] & a[1] & a[0];

endmodule

Reduction Operators: Schematic View

ANDa[7:0]
[7:0]

[0]

[1]

[2]

[3]

[4]

[5]

[6]

[7]

y

8-input AND gate



Conditional Assignment in Behavioral Verilog

n ? :  is also called a ternary operator as it operates on three 
inputs:
q s
q d1
q d0

module mux2(input  [3:0] d0, d1, 
            input        s,
            output [3:0] y);

   assign y = s ? d1 : d0; 
   // if (s) then y=d1 else y=d0;

endmodule

Conditional Assignment: Schematic View

More Complex Conditional Assignments
module mux4(input  [3:0] d0, d1, d2, d3 
            input [1:0] s,
            output [3:0] y);

   assign y = s[1] ? ( s[0] ? d3 : d2)
                   : ( s[0] ? d1 : d0); 
  // if (s1) then 
  //      if (s0) then y=d3 else y=d2
  // else
  //      if (s0) then y=d1 else y=d0

endmodule

Even More Complex Conditional Assignments

module mux4(input  [3:0] d0, d1, d2, d3 
            input [1:0] s,
            output [3:0] y);

   assign y = (s == 2’b11) ? d3 :
              (s == 2’b10) ? d2 :
              (s == 2’b01) ? d1 :
              d0;
// if      (s = “11” ) then y= d3
// else if (s = “10” ) then y= d2
// else if (s = “01” ) then y= d1
// else                     y= d0

endmodule

Precedence of Operations in Verilog

Highest

Lowest

How to Express Numbers?

N’Bxx
8’b0000_0001

n (N) Number of bits
q Expresses how many bits will be used to store the value

n (B) Base
q Can be b (binary), h (hexadecimal), d (decimal), o (octal)

n (xx) Number
q The value expressed in base
q Can also have X (invalid) and Z (floating), as values
q Underscore _ can be used to improve readability

Number Representation in Verilog

Verilog Stored Number Verilog Stored Number

4’b1001 1001 4’d5 0101

8’b1001 0000 1001 12’hFA3 1111 1010 0011

8’b0000_1001 0000 1001 8’o12 00 001 010

8’bxX0X1zZ1 XX0X 1ZZ1 4’h7 0111

‘b01 0000 .. 0001 12’h0 0000 0000 0000

50

32 bits 
(default)



Reminder: Floating Signals (Z)

module tristate_buffer(input [3:0] a, 
input en, 
output [3:0] y);

assign y = en ? a : 4'bz;

endmodule

y_1[3:0]

y[3:0][3:0]
en

a[3:0] [3:0] [3:0][3:0]

51

n Floating signal: Signal that is not driven by any circuit
q Open circuit, floating wire

n Also known as: high impedance, hi-Z, tri-stated signals



Recall: Tri-State Buffer
n A tri-state buffer enables gating of different signals onto a 

wire

n Floating signal (Z): Signal that is not driven by any circuit
q Open circuit, floating wire

52

A tri-state buffer
acts like a switch



Recall: Example: Use of Tri-State Buffers
n Imagine a wire connecting the CPU and memory

q At any time only the CPU or the memory can place a value on 
the wire, both not both

q You can have two tri-state buffers: one driven by CPU, the 
other memory; and ensure at most one is enabled at any time

Recall: Example Design with Tri-State Buffers

CPU

Memory

GateMem

GateCPU

Shared Bus



Recall: Another Example

Truth Table for AND Gate with Z and X

AND A
0 1 Z X

B

0 0 0 0 0

1 0 1 X X

Z 0 X X X

X 0 X X X

Recall: Simplified Priority Circuit
n Priority Circuit

q Inputs: “Requestors” with priority levels
q Outputs: “Grant” signal for each requestor
q Example 4-bit priority circuit

57

X (Don’t Care) means I don’t care what the value of this input is



What Happens with HDL Code?
n Synthesis (i.e., Hardware Synthesis)

q Modern tools are able to map synthesizable HDL code into 
low-level cell libraries à netlist describing gates and wires

q They can perform many optimizations
q … however they can not guarantee that a solution is optimal

n Mainly due to computationally expensive placement and routing
algorithms

n Need to describe your circuit in HDL in a nice-to-synthesize way
q Most common way of Digital Design these days

n Simulation
q Allows the behavior of the circuit to be verified without actually 

manufacturing the circuit
q Simulators can work on structural or behavioral HDL
q Simulation is essential for functional and timing verification

Recall This “example”

module example (a, b, c, y);
 input a;
 input b; 
 input c;
 output y;

// here comes the circuit description
assign y = ~a & ~b & ~c |
            a & ~b & ~c |
            a & ~b &  c;   

endmodule

Synthesizing the “example”

60

ANDa

y
b

c

AND

AND

OR



Simulating the “example”

time

si
gn

al
s

Waveform Diagram

1
1
1
0

ANDa

yb

c

AND

AND

OR



A Note on Hardware Synthesis

62Read H&H Chapter 4.1



What We Have Seen So Far
n Describing structural hierarchy with Verilog

q Instantiate modules in an other module
n Describing functionality using behavioral modeling

n Writing simple logic equations
q We can write AND, OR, XOR, … 

n Multiplexer functionality
q If … then … else

n We can describe constants

n But there is more...

More Verilog Examples

n We can write Verilog code in many different ways

n Let’s see how we can express the same functionality by 
developing Verilog code

q At a low-level of abstraction
n Poor readability
n Easier automated optimization (especially for low-level tools)

q At a high-level of abstraction
n Better readability
n More difficult automated optimization (large search space)

Comparing Two Numbers
n Defining your own gates as new modules

n We will use our gates to show different ways of 
implementing a 4-bit comparator (equality checker)

module MyXnor (input A, B,
               output Z);

   assign Z = ~(A ^ B); //not XOR

endmodule

module MyAnd (input A, B,
              output Z);

   assign Z = A & B;    // AND

endmodule

A 2-input XNOR gate A 2-input AND gate

Recall: Equality Checker (Compare if Equal)
n Checks if two N-input values are exactly the same
n Example: 4-bit Comparator



Gate-Level Implementation

module compare (input a0, a1, a2, a3, b0, b1, b2, b3,
                output eq);
       wire c0, c1, c2, c3, c01, c23;

MyXnor i0 (.A(a0), .B(b0), .Z(c0) ); // XNOR
MyXnor i1 (.A(a1), .B(b1), .Z(c1) ); // XNOR
MyXnor i2 (.A(a2), .B(b2), .Z(c2) ); // XNOR
MyXnor i3 (.A(a3), .B(b3), .Z(c3) ); // XNOR
MyAnd haha (.A(c0), .B(c1), .Z(c01) ); // AND
MyAnd hoho (.A(c2), .B(c3), .Z(c23) ); // AND
MyAnd bubu (.A(c01), .B(c23), .Z(eq) ); // AND

endmodule

Using Logical Operators

module compare (input a0, a1, a2, a3, b0, b1, b2, b3,
                output eq);
       wire c0, c1, c2, c3, c01, c23;

MyXnor i0 (.A(a0), .B(b0), .Z(c0) ); // XNOR
MyXnor i1 (.A(a1), .B(b1), .Z(c1) ); // XNOR
MyXnor i2 (.A(a2), .B(b2), .Z(c2) ); // XNOR
MyXnor i3 (.A(a3), .B(b3), .Z(c3) ); // XNOR
assign c01 = c0 & c1;
assign c23 = c2 & c3;
assign eq  = c01 & c23;

endmodule

Eliminating Intermediate Signals

module compare (input a0, a1, a2, a3, b0, b1, b2, b3,
                output eq);
       wire c0, c1, c2, c3;

MyXnor i0 (.A(a0), .B(b0), .Z(c0) ); // XNOR
MyXnor i1 (.A(a1), .B(b1), .Z(c1) ); // XNOR
MyXnor i2 (.A(a2), .B(b2), .Z(c2) ); // XNOR
MyXnor i3 (.A(a3), .B(b3), .Z(c3) ); // XNOR
// assign c01 = c0 & c1;
// assign c23 = c2 & c3;
// assign eq  = c01 & c23;
assign eq  = c0 & c1 & c2 & c3;

endmodule

69



Multi-Bit Signals (Bus)

module compare (input [3:0] a, input [3:0] b,
                output eq);
       wire [3:0] c; // bus definition

MyXnor i0 (.A(a[0]), .B(b[0]), .Z(c[0]) ); // XNOR
MyXnor i1 (.A(a[1]), .B(b[1]), .Z(c[1]) ); // XNOR
MyXnor i2 (.A(a[2]), .B(b[2]), .Z(c[2]) ); // XNOR
MyXnor i3 (.A(a[3]), .B(b[3]), .Z(c[3]) ); // XNOR

assign eq  = &c; // short format

endmodule

Bitwise Operations

module compare (input [3:0] a, input [3:0] b,
                output eq);
       wire [3:0] c; // bus definition

// MyXnor i0 (.A(a[0]), .B(b[0]), .Z(c[0]) );
// MyXnor i1 (.A(a[1]), .B(b[1]), .Z(c[1]) );
// MyXnor i2 (.A(a[2]), .B(b[2]), .Z(c[2]) );
// MyXnor i3 (.A(a[3]), .B(b[3]), .Z(c[3]) );

assign c = ~(a ^ b); // XNOR

assign eq  = &c; // short format

endmodule

Highest Abstraction Level: Comparing Two Numbers

module compare (input [3:0] a, input [3:0] b,
                output eq);

// assign c = ~(a ^ b); // XNOR

// assign eq  = &c; // short format

assign eq = (a == b) ? 1 : 0; // really short

endmodule

Writing More Reusable Verilog Code
n We have a module that can compare two 4-bit numbers

n What if in the overall design we need to compare:
q 5-bit numbers?
q 6-bit numbers?
q …
q N-bit numbers?
q Writing code for each case looks tedious

n What could be a better way?

Parameterized Modules

We can set the parameters to different values 
when instantiating the module

module mux2
  #(parameter width = 8)  // name and default value
   (input  [width-1:0] d0, d1, 
    input              s,
    output [width-1:0] y);

  assign y = s ? d1 : d0; 
endmodule 

74

In Verilog, we can define module parameters



Instantiating Parameterized Modules

// If the parameter is not given, the default (8) is assumed
   mux2 i_mux (d0, d1, s, out);

// The same module with 12-bit bus width:
   mux2 #(12) i_mux_b (d0, d1, s, out);

// A more verbose version:
   mux2 #(.width(12)) i_mux_b (.d0(d0), .d1(d1),
                               .s(s), .out(out));
 

module mux2
  #(parameter width = 8)  // name and default value
   (input  [width-1:0] d0, d1, 
    input              s,
    output [width-1:0] y);

  assign y = s ? d1 : d0; 
endmodule 

What About Timing?
n It is possible to define timing relations in Verilog. BUT:

q These are ONLY for simulation
q They CAN NOT be synthesized
q They are used for modeling delays in a circuit

‘timescale 1ns/1ps
module simple (input a, output z1, z2);

assign #5 z1 = ~a; // inverted output after 5ns
assign #9 z2 = a;  // output after 9ns

endmodule

More on this soon



Good Practices
n Develop/use a consistent naming style

n Use MSB to LSB ordering for buses
q Use “a[31:0]”, not “a[0:31]”

n Define one module per file
q Makes managing your design hierarchy easier

n Use a file name that matches module name
q e.g., module TryThis is defined in a file called TryThis.v

n Always keep in mind that Verilog describes hardware

78



Summary (HDL for Combinational Logic)

n We have seen an overview of Verilog

n Discussed structural and behavioral modeling

n Studied combinational logic constructs

We Covered 
Until This Point

in Lecture

Slides for Future Lectures

Implementing Sequential Logic 
Using Verilog

Sequential = Combinational + Memory

Sequential Logic in Verilog
n Define blocks that have memory

q Flip-Flops, Latches, Finite State Machines

n Sequential Logic state transition is triggered by a “CLOCK” 
signal  
q Latches are sensitive to level of the signal
q Flip-flops are sensitive to the transitioning of signal

n Combinational HDL constructs are not sufficient to express 
sequential logic
q We need new constructs:

n always
n posedge/negedge

The “always” Block

always @ (sensitivity list)
statement;

Whenever the event in the sensitivity list occurs, 
the statement is executed



n 1) state change on clock edge, 2) data available for full cycle

D Latch (Second)
D Latch (First)

Recall: The D Flip-Flop

D
Q

CLK

n When the clock is low, 1st latch propagates D to the input of the 2nd (Q unchanged)
n Only when the clock is high, 2nd latch latches D (Q stores D)

q At the rising edge of clock (clock going from 0->1), Q gets assigned D

CLK:
0
1



Recall: The D Flip-Flop
n 1) state change on clock edge, 2) data available for full cycle

n At the rising edge of clock (clock going from 0->1), Q gets assigned D
n At all other times, Q is unchanged

CLK:
0
1

CLKD Q

Q
__

D Flip-Flop



Recall: The D Flip-Flop
n 1) state change on clock edge, 2) data available for full cycle

88

n At the rising edge of clock (clock going from 0->1), Q gets assigned D
n At all other times, Q is unchanged

CLK:
0
1

CLKD Q

Q
__

D Flip-Flop
We can use D Flip-Flops 

to implement the state register

Example: D Flip-Flop

module flop(input clk, 
input [3:0] d, 
output reg [3:0] q);

always @ (posedge clk)
q <= d;                // pronounced “q gets d”

endmodule

n posedge defines a rising edge (transition from 0 to 1). 

n Statement executed when the clk signal rises (posedge of clk) 

n Once the clk signal rises: the value of d is copied to q

Example: D Flip-Flop

module flop(input clk, 
input [3:0] d, 
output reg [3:0] q);

always @ (posedge clk)
q <= d;                // pronounced “q gets d”

endmodule

n assign statement is not used within an always block
n <= describes a non-blocking assignment

q We will see the difference between blocking assignment and 
non-blocking assignment soon

Example: D Flip-Flop

module flop(input clk, 
input [3:0] d, 
output reg [3:0] q);

always @ (posedge clk)
q <= d;                // pronounced “q gets d”

endmodule

n Assigned variables need to be declared as reg
n The name reg does not necessarily mean that the value is a 

register (It could be, but it does not have to be)
n We will see examples later



Asynchronous and Synchronous Reset
n Reset signals are used to initialize the hardware to a known 

state
q Usually activated at system start (on power up)

n Asynchronous Reset
q The reset signal is sampled independent of the clock
q Reset gets the highest priority
q Sensitive to glitches, may have metastability issues

n Will be discussed in the Timing & Verification Lecture 

n Synchronous Reset
q The reset signal is sampled with respect to the clock
q The reset should be active long enough to get sampled at the 

clock edge
q Results in completely synchronous circuit

Recall: Asynchronous vs. Synchronous State Changes

n Sequential lock we saw is an asynchronous “machine”
q State transitions occur when they occur
q There is nothing that synchronizes when each state transition 

must occur 

n Most modern computers are synchronous “machines”
q State transitions take place after fixed units of time
q Controlled in part by a clock, as we will see soon

n These are two different design paradigms, with tradeoffs
q Synchronous control can be easier to get correct when the 

system consists of many components and many states
q Asynchronous control can be more efficient (no clock overheads)

93We will assume synchronous systems in most of this course

D Flip-Flop with Asynchronous Reset

module flop_ar (input            clk,
input            reset, 
input      [3:0] d, 
output reg [3:0] q);

always @ (posedge clk, negedge reset)
begin

if (reset == 0) q <= 0;   // when reset
else            q <= d;   // when clk

end
endmodule

n In this example: two events can trigger the process:
q A rising edge on clk
q A falling edge on reset

D Flip-Flop with Asynchronous Reset

module flop_ar (input            clk,
input            reset, 
input      [3:0] d, 
output reg [3:0] q);

always @ (posedge clk, negedge reset)
begin

if (reset == 0) q <= 0;   // when reset
else            q <= d;   // when clk

end
endmodule

n For longer statements, a begin-end pair can be used
q To improve readability
q In this example, it was not necessary, but it is a good idea

D Flip-Flop with Asynchronous Reset

module flop_ar (input            clk,
input            reset, 
input      [3:0] d, 
output reg [3:0] q);

always @ (posedge clk, negedge reset)
begin

if (reset == 0) q <= 0;   // when reset
else            q <= d;   // when clk

end
endmodule

n First reset is checked: if reset is 0, q is set to 0.
q This is an asynchronous reset as the reset can happen 

independently of the clock (on the negative edge of reset signal)
n If there is no reset, then regular assignment takes effect

D Flip-Flop with Synchronous Reset

module flop_sr (input            clk,
input            reset, 
input      [3:0] d, 
output reg [3:0] q);

always @ (posedge clk)
begin

if (reset == ‘0’) q <= 0;   // when reset
else              q <= d;   // when clk

end
endmodule

n The process is sensitive to only clock
q Reset happens only when the clock rises. This is a 

synchronous reset

D Flip-Flop with Enable and Reset

module flop_en_ar (input            clk,
input            reset,
input            en,
input      [3:0] d, 
output reg [3:0] q);

always @ (posedge clk, negedge reset)
begin

if (reset == ‘0’) q <= 0;   // when reset
else if (en)      q <= d;   // when en AND clk

end
endmodule

n A flip-flop with enable and reset
q Note that the en signal is not in the sensitivity list

n q gets d only when clk is rising and en is 1

Example: D Latch

module latch (input            clk, 
input      [3:0] d, 
output reg [3:0] q);

always @ (clk, d)
if (clk) q <= d;      // latch is transparent when

// clock is 1
endmodule



Summary: Sequential Statements So Far
n Sequential statements are within an always block

n The sequential block is triggered with a change in the 
sensitivity list

n Signals assigned within an always must be declared as reg

n We use <= for (non-blocking) assignments and do not use 
assign within the always block.

Basics of always Blocks
module example (input            clk, 

input      [3:0] d, 
output reg [3:0] q);

wire [3:0] normal;         // standard wire
reg [3:0] special; // assigned in always

always @ (posedge clk)
special <= d;            // first FF array

assign normal = ~special; // simple assignment

always @ (posedge clk)
q <= normal;             // second FF array

endmodule

You can have as many always blocks as needed
Assignment to the same signal in different always blocks is not allowed!

Why Does an always Block Remember?

module flop (input            clk,
input      [3:0] d, 
output reg [3:0] q);

always @ (posedge clk)
begin

q <= d; // when clk rises copy d to q
end

endmodule

n This statement describes what happens to signal q
n … but what happens when the clock is not rising?
n The value of q is preserved (remembered)

An always Block Does NOT Always Remember

module comb (input            inv,
input      [3:0] data, 
output reg [3:0] result);

always @ (inv, data)       // trigger with inv, data
if (inv) result <= ~data;// result is inverted data
else result <= data; // result is data

endmodule

n This statement describes what happens to signal result
q When inv is 1, result is ~data
q When inv is not 1, result is data 

n The circuit is combinational (no memory)
q result is assigned a value whenever an input value changes & in all 

cases of the if .. else block



always Blocks for Combinational Circuits
n An always block defines combinational logic if:

q All outputs are always (continuously) updated
1. All right-hand side signals are in the sensitivity list

n You can use always @* for short
2. All left-hand side signals get assigned in every possible condition 
of if .. else and case blocks

n It is easy to make mistakes and unintentionally describe 
memorizing elements (latches)
q Vivado will most likely warn you. Make sure you check the 

warning messages

n Always blocks allow powerful combinational logic statements
q if .. else
q case

Sequential or Combinational?

wire enable, data;
reg out_a, out_b;

always @ (*) begin
out_a = 1’b0;
if(enable) begin

out_a = data;
out_b = data;

end
end

Sequential

wire enable, data;
reg out_a, out_b; 

always @ (data) begin
 out_a = 1’b0;
 out_b = 1’b0;
 if(enable) begin
  out_a = data;
  out_b = data;
 end
end

Sequential

No assignment for ~enable Not in the sensitivity list

The always Block is NOT Always Practical/Nice

reg [31:0] result;
wire [31:0] a, b, comb;
wire sel,

always @ (a, b, sel)    // trigger with a, b, sel
if (sel) result <= a; // result is a
else result <= b; // result is b

assign comb = sel ? a : b;

n Both statements describe the same multiplexer

n In this case, the always block is more work

always Block for Case Statements (Handy!)
module sevensegment (input [3:0] data, 

output reg [6:0] segments);

always @ ( * )                  // * is short for all signals
case (data)                   // case statement 

4'd0: segments = 7'b111_1110;  // when data is 0
4'd1: segments = 7'b011_0000;  // when data is 1 
4'd2: segments = 7'b110_1101;
4'd3: segments = 7'b111_1001;
4'd4: segments = 7'b011_0011;
4'd5: segments = 7'b101_1011;
// etc etc
default: segments = 7'b000_0000; // required

endcase

endmodule



Summary: always Block
n if .. else can only be used in always blocks

n The always block is combinational only if all regs within the 
block are always assigned to a signal
q Use the default case to make sure you do not forget an 

unimplemented case, which may otherwise result in a latch

n Use casex statement to be able to check for don’t cares

Non-Blocking and Blocking Assignments

always @ (a)
begin

a <= 2’b01;
b <= a;

// all assignments are made here
// b is not (yet) 2’b01
end

always @ (a)
begin
   a = 2’b01;
// a is 2’b01
   b = a;
// b is now 2’b01 as well
end

Non-blocking (<=) Blocking (=)

n All assignments are made 
at the end of the block

n All assignments are made 
in parallel, process flow is
not-blocked

n Each assignment is made 
immediately 

n Process waits until the first 
assignment is complete, it 
blocks progress

n Similar to sequential programs



Why Use (Non)-Blocking Statements
n Non-blocking statements allow operating on “old” values

q Enable easy sequential logic descriptions

n Blocking statements allow a sequence of operations
q Allow operating on immediately updated values
q More like a “software” programming language

n If the sensitivity list is correct, a block with non-blocking 
statements will eventually evaluate to the same result as 
the same block with blocking statements
q This may require some additional iterations 

Example: Blocking Assignment

always @ ( * )
begin
p    = a ^ b ;         // p    = 0 
g    = a & b ;         // g    = 0
s    = p ^ cin ;       // s    = 0 
cout = g | (p & cin) ; // cout = 0

end

n Assume all inputs are initially ‘0’

n If a changes to ‘1’
q All values are updated in order

1
0
1
0

The Same Example: Non-Blocking Assignment

always @ ( * )
begin
p    <= a ^ b ;         // p    = 0 
g    <= a & b ;         // g    = 0
s    <= p ^ cin ;       // s    = 0 
cout <= g | (p & cin) ; // cout = 0

end

n Assume all inputs are initially ‘0’

n If a changes to ‘1’
q All assignments are concurrent
q When s is being assigned, p is still 0

1
0
0
0

The Same Example: Non-Blocking Assignment

always @ ( * )
begin
p    <= a ^ b ;         // p    = 1 
g    <= a & b ;         // g    = 0
s    <= p ^ cin ;       // s    = 0 
cout <= g | (p & cin) ; // cout = 0

end

n After the first iteration, p has changed to ‘1’ as well

n Since there is a change in p, the process triggers again
n This time s is calculated with p=1

1
0
1
0

Rules for Signal Assignment

n Use always @(posedge clk) and non-blocking
assignments (<=) to model synchronous sequential logic

n Use continuous assignments (assign) to model simple 
combinational logic

always @ (posedge clk)
   q <= d; // non-blocking

assign y = a & b;



Rules for Signal Assignment (Cont.)

n Use always @ (*) and blocking assignments (=) to model 
more complicated combinational logic

n You cannot make assignments to the same signal in more 
than one always block or in a continuous assignment

always @ (*)
   a = b;

always @ (*)
   a = c;

always @ (*)
   a = b;

assign a = c;



Recall: Finite State Machines (FSMs)
n Each FSM consists of three separate parts:

q next state logic
q state register
q output logic

116

CLK
M Nk knext

state
logic

output
logic

inputs outputsstate
next
state

state register

At the beginning of the clock cycle, next state is latched into the state register



Recall: Finite State Machines (FSMs) Consist of:

n Sequential Circuits
q State register(s)

n Store the current state and 
n Load the next state at the clock edge

n Combinational Circuits
q Next state logic

n Determines what the next state will be

q Output logic
n Generates the outputs

FSM Example 1: Divide the Clock Frequency by 3 

The output Y is HIGH for one clock cycle out of every 3. In other 
words, the output divides the frequency of the clock by 3. 

Implementing FSM Example 1: Definitions

module divideby3FSM (input clk, 
input reset, 
output q);

reg [1:0] state, nextstate;

parameter S0 = 2'b00;
parameter S1 = 2'b01;
parameter S2 = 2'b10;

n We define state and nextstate as 2-bit reg
n The parameter descriptions are optional, it makes reading 

easier

Implementing FSM Example 1: State Register

// state register
always @ (posedge clk, posedge reset)

if (reset) state <= S0;
else       state <= nextstate;

n This part defines the state register (memorizing process)
n Sensitive to only clk, reset
n In this example, reset is active when it is ‘1’ (active-high)

Next
State

Current
State

S’ S

CLK

Implementing FSM Example 1: Next State Logic

// next state logic
   always @ (*)
      case (state)
         S0:      nextstate = S1;
         S1:      nextstate = S2;
         S2:      nextstate = S0;
         default: nextstate = S0;
      endcase

CLK
M Nk knext

state
logic

output
logic

inputs outputsstate
next
state

Implementing FSM Example 1: Output Logic

// output logic
   assign q = (state == S0);

n In this example, output depends only on state
q Moore type FSM

CLK
M Nk knext

state
logic

output

next
state

Implementation of FSM Example 1
module divideby3FSM (input clk, input reset, output q);
   reg  [1:0] state, nextstate;
   
   parameter S0 = 2'b00; parameter S1 = 2'b01; parameter S2 = 2'b10;
   
   always @ (posedge clk, posedge reset) // state register
      if (reset) state <= S0;
      else       state <= nextstate;
   
   always @ (*)                          // next state logic
      case (state)
         S0:      nextstate = S1;
         S1:      nextstate = S2;
         S2:      nextstate = S0;
         default: nextstate = S0;
      endcase
   assign q = (state == S0);            // output logic
endmodule 



FSM Example 2: Recall the Smiling Snail
n Alyssa P. Hacker has a snail that crawls down a paper tape 

with 1’s and 0’s on it 
n The snail smiles whenever the last four digits it has crawled 

over are 1101
n Design Moore and Mealy FSMs of the snail’s brain

Moore

Mealy

Implementing FSM Example 2: Definitions
module SmilingSnail (input clk, 
                     input reset,
                     input number,  
                     output smile);

   reg  [1:0] state, nextstate;
   
   parameter S0 = 2'b00;
   parameter S1 = 2'b01;
   parameter S2 = 2'b10;
   parameter S3 = 2’b11;
   

digit/smile

Implementing FSM Example 2: State Register

// state register
   always @ (posedge clk, posedge reset)
      if (reset) state <= S0;
      else       state <= nextstate;

n This part defines the state register (memorizing process)

n Sensitive to only clk, reset

n In this example reset is active when ‘1’ (active-high)

Implementing FSM Example 2: Next State Logic

// next state logic
   always @ (*)
      case (state)
         S0: if (number) nextstate = S1;
             else   nextstate = S0;
         S1: if (number) nextstate = S2;
             else   nextstate = S0;
         S2: if (number) nextstate = S2;
             else   nextstate = S3;
         S3: if (number) nextstate = S1;
             else   nextstate = S0;
         default:   nextstate = S0;
      endcase

Implementing FSM Example 2: Output Logic

// output logic
   assign smile = (number & state == S3);

n In this example, output depends on state and input
q Mealy type FSM

n We used a simple combinational assignment

Implementation of FSM Example 2
module SmilingSnail (input clk, 
                     input reset,
                     input number,
                     output smile);

   reg  [1:0] state, nextstate;
   
   parameter S0 = 2'b00;
   parameter S1 = 2'b01;
   parameter S2 = 2'b10;
   parameter S3 = 2’b11;

   // state register
   always @ (posedge clk, posedge 
reset)
      if (reset) state <= S0;
      else      state <= nextstate;   

always @ (*) // next state logic
      case (state)
         S0: if (number)
                  nextstate = S1;
             else nextstate = S0; 
         S1: if (number)
                  nextstate = S2;
             else nextstate = S0;
         S2: if (number)
                  nextstate = S2;
             else nextstate = S3;
         S3: if (number)
                  nextstate = S1;
             else nextstate = S0;
         default: nextstate = S0;
      endcase
   // output logic
assign smile = (number & state==S3);

endmodule



What Did We Learn?
n Basics of describing sequential circuits in Verilog

n The always statement
q Needed for describing memorizing elements (flip-flops, latches)
q Can also be used to describe combinational circuits

n Blocking vs Non-blocking statements
q = assigns the value immediately
q <= assigns the value at the end of the block

n Describing FSMs in Verilog
q Next state logic
q State assignment
q Output logic

130



Next Lecture: 
Timing and Verification

